@ECHO off
@REM =======================================================
@REM this script uses itself as data input
@REM the data format is: REM <app_folder> <1/0>
@REM the data line MUST begin with "REM", then follows
@REM a space char, then follows the application folder
@REM name, and then, a flag indicating YES or NO (1=YES)
@REM when calling this script, make sure `pfbak.cmd' is
@REM callable.
@REM ---- copyright (C) Xia Zuoling, 2009
@REM ---- updated: 17:03 2009-8-21 CST
@REM =======================================================

@REM Backup target list

REM CTerm 1
REM EditPlus 1
REM firefox 0
REM FtpRush 1
REM GreenBrowser 1
REM HyperSnap 1
REM IrfanView 0
REM Thunder 0
REM totalcmd 1

@REM main program

for /F "tokens=1,2,3" %%i in (%~0) do (
	if "%%~i"=="REM" if "%%~k"=="1" (
		echo Backing : %%j
		call pfbak.cmd %%j
	) else (
		echo Skipping: %%j
	)
)
